Pensieve header: Implementing U(g0).

Reminders



Global Mathematica Initialization

```
FileNameJoin[{$BaseDirectory, "Kernel", "init.m"}]
```

```
C:\ProgramData\Mathematica\Kernel\init.m
```

In the above file, I have (among other things):

```
If[$FrontEnd =!= Null,
 SetOptions[$FrontEnd,
  InputAliases → {"Zhe" → "ж", "zhe" → "ж", "<<" → "≺", "<<=" → "≼", ">>" → "≻", ">>=" → "≽"}
 ]
]
```

Implementing g_0

```
PBWRule = {e → 1, l → 2, f → 3};
B[U@l, U@e] = -(B[U@e, U@l] = U@e);
B[U@f, U@l] = -(B[U@l, U@f] = U@f);
B[U@e, U@f] = -(B[U@f, U@e] = h U[]);
```

```
Uᵢ_[ℰ_] := ℰ /. {h → hᵢ, t → tᵢ, u_U :→ Replace[u, x_ :→ xᵢ, 1]};
B[x_, x_] = 0;
B[U[(x_)ᵢ_], U[(y_)ᵢ_]] := B[U[xᵢ], U[yᵢ]] = Uᵢ[B[U@x, U@y]];
B[U[(x_)ᵢ_], U[(y_)ⱼ_]] /; i =!= j := 0;
B[x_, y_] := x ** y - y ** x;
```

```
x_ ≼ y_ := OrderedQ[{x, y} /. PBWRule];
x_ ≺ y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] := If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[];
UU[l_, r___] := U[l] ** UU[r];
```

Testing g_0

```
UProducts[{}, 0] = {UU[]};
UProducts[{}, n_Integer] /; n > 0 = {};
UProducts[{x_, xs___}, n_Integer] :=
  Sort@Flatten@Table[UU[x^k] ** u, {k, 0, n}, {u, UProducts[{xs}, n - k]}];
UProducts[xs_List, k_Integer, n_Integer] := UProducts[Flatten@Table[x_j, {x, xs}, {j, k}], n];
UProducts[any__, {n_}] := Flatten@Table[UProducts[any, k], {k, 0, n}];
```

```
B[U@f1, U@e1]
```

$h_1\, U[\,]$

```
UProducts[{e, l, f}, 2, {3}]
```

$\{U[\,], U[e_1], U[e_2], U[f_1], U[f_2], U[l_1], U[l_2], U[e_1, e_1], U[e_1, e_2], U[e_1, f_1], U[e_1, f_2],$
$U[e_1, l_1], U[e_1, l_2], U[e_2, e_2], U[e_2, f_1], U[e_2, f_2], U[e_2, l_1], U[e_2, l_2], U[f_1, f_1], U[f_1, f_2],$
$U[f_2, f_2], U[l_1, f_1], U[l_1, f_2], U[l_1, l_1], U[l_1, l_2], U[l_2, f_1], U[l_2, f_2], U[l_2, l_2], U[e_1, e_1, e_1],$
$U[e_1, e_1, e_2], U[e_1, e_1, f_1], U[e_1, e_1, f_2], U[e_1, e_1, l_1], U[e_1, e_1, l_2], U[e_1, e_2, e_2],$
$U[e_1, e_2, f_1], U[e_1, e_2, f_2], U[e_1, e_2, l_1], U[e_1, e_2, l_2], U[e_1, f_1, f_1], U[e_1, f_1, f_2], U[e_1, f_2, f_2],$
$U[e_1, l_1, f_1], U[e_1, l_1, f_2], U[e_1, l_1, l_1], U[e_1, l_1, l_2], U[e_1, l_2, f_1], U[e_1, l_2, f_2], U[e_1, l_2, l_2],$
$U[e_2, e_2, e_2], U[e_2, e_2, f_1], U[e_2, e_2, f_2], U[e_2, e_2, l_1], U[e_2, e_2, l_2], U[e_2, f_1, f_1], U[e_2, f_1, f_2],$
$U[e_2, f_2, f_2], U[e_2, l_1, f_1], U[e_2, l_1, f_2], U[e_2, l_1, l_1], U[e_2, l_1, l_2], U[e_2, l_2, f_1], U[e_2, l_2, f_2],$
$U[e_2, l_2, l_2], U[f_1, f_1, f_1], U[f_1, f_1, f_2], U[f_1, f_2, f_2], U[f_2, f_2, f_2], U[l_1, f_1, f_1], U[l_1, f_1, f_2],$
$U[l_1, f_2, f_2], U[l_1, l_1, f_1], U[l_1, l_1, f_2], U[l_1, l_1, l_1], U[l_1, l_1, l_2], U[l_1, l_2, f_1], U[l_1, l_2, f_2],$
$U[l_1, l_2, l_2], U[l_2, f_1, f_1], U[l_2, f_1, f_2], U[l_2, f_2, f_2], U[l_2, l_2, f_1], U[l_2, l_2, f_2], U[l_2, l_2, l_2]\}$

```
bas = UProducts[{e, l, f}, 2, {3}];
Table[B[x, y] + B[y, x], {x, bas}, {y, bas}] // Flatten // Union
```

$\{0\}$

```
bas = UProducts[{e, l, f}, 2, {2}];
Table[
   {x, y, z} = xyz;
   Simp[B[B[x, y], z] + B[B[y, z], x] + B[B[z, x], y]],
   {xyz, Subsets[bas, {3}]}
  ] // Flatten // Union
```

$\{0\}$

```
bas = UProducts[{e, l, f}, 2, {2}];
Table[
   {x, y, z} = xyz;
   Simp[x ** (y ** z) - (x ** y) ** z],
   {xyz, Subsets[bas, {3}]}
  ] // Flatten // Union
```

```
{0}
```

Testing CYBE

```
r_i_,j_ := h_i UU[l_j] + UU[f_i, e_j]
```

```
B[r_1,2, r_1,3]
```

```
0
```

```
B[r_1,3, r_2,3]
```

```
h_2 U[e_3, f_1] - h_1 U[e_3, f_2]
```

```
B[r_1,2, r_2,3]
```

```
-h_2 U[e_3, f_1] + h_1 U[e_3, f_2]
```

```
B[r_1,2, r_1,3] + B[r_1,3, r_2,3] + B[r_1,2, r_2,3]
```

```
0
```

Testing YBE

```
UExp[n_Integer, u_] := Module[{t},
   t = U[];
   Simp[t + Sum[t = t ** u / k!, {k, n}]]
  ];
R_i_,j_[n_] := UExp[n, r_i,j];
```

```
UExp[5, U@e_1]
```

$$U[] + U[e_1] + \frac{1}{2} U[e_1, e_1] + \frac{1}{6} U[e_1, e_1, e_1] + \frac{1}{24} U[e_1, e_1, e_1, e_1] + \frac{1}{120} U[e_1, e_1, e_1, e_1, e_1]$$

```
R_1,2[4]
```

$$U[] + h_1 U[l_2] + \left(1 - \frac{h_1}{2} + \frac{h_1^2}{6} - \frac{h_1^3}{24}\right) U[e_2, f_1] + \frac{1}{2} h_1^2 U[l_2, l_2] + \left(h_1 - \frac{h_1^2}{2} + \frac{h_1^3}{6}\right) U[e_2, l_2, f_1] +$$
$$\frac{1}{6} h_1^3 U[l_2, l_2, l_2] + \left(\frac{1}{2} - \frac{h_1}{2} + \frac{7 h_1^2}{24}\right) U[e_2, e_2, f_1, f_1] + \left(\frac{h_1^2}{2} - \frac{h_1^3}{4}\right) U[e_2, l_2, l_2, f_1] + \frac{1}{24} h_1^4 U[l_2, l_2, l_2, l_2] +$$
$$\left(\frac{h_1}{2} - \frac{h_1^2}{2}\right) U[e_2, e_2, l_2, f_1, f_1] + \frac{1}{6} h_1^3 U[e_2, l_2, l_2, l_2, f_1] + \left(\frac{1}{6} - \frac{h_1}{4}\right) U[e_2, e_2, e_2, f_1, f_1, f_1] +$$
$$\frac{1}{4} h_1^2 U[e_2, e_2, l_2, l_2, f_1, f_1] + \frac{1}{6} h_1 U[e_2, e_2, e_2, l_2, f_1, f_1, f_1] + \frac{1}{24} U[e_2, e_2, e_2, e_2, f_1, f_1, f_1, f_1]$$

```
With[{n = 2}, Simp[R_1,2[n] ** R_1,3[n] ** R_2,3[n] - R_2,3[n] ** R_1,3[n] ** R_1,2[n]]] // Short
```

$$\left(-\frac{1}{2} h_1^2 h_2 + \frac{1}{4} h_1^2 h_2^2\right) U[e_3, l_2, f_1] + \left(\frac{h_1^3}{2} - \frac{1}{4} h_1^3 h_2\right) U[e_3, l_2, f_2] +$$
$$\text{<<130>>} + \frac{1}{2} h_1 U[e_2, e_3, e_3, e_3, e_3, f_1, f_1, f_1, f_2, f_2]$$

```
ToDegree[n_][ℰ_] :=
 Simp[ℰ /. {h_i_ :> ħ h_i, u_U :> ħ^Count[u,f_] u}] /. a_. x_U :> Normal[Series[a, {ħ, 0, n}]] * x /. ħ → 1
```

```
With[{n = 2}, Simp[R1,2[n] ** R1,3[n] ** R2,3[n] - R2,3[n] ** R1,3[n] ** R1,2[n]] // ToDegree[n + 2]]
```

$$-\frac{1}{2} h_1^2 h_2 U[e_3, l_2, f_1] + \frac{1}{2} h_1^3 U[e_3, l_2, f_2] + \frac{1}{2} h_1^2 h_2 U[e_3, l_3, f_1] - \frac{1}{2} h_1^3 U[e_3, l_3, f_2] - h_1 h_2 U[e_2, e_3, f_1, f_1] + h_1^2 U[e_2, e_3, f_1, f_2] + h_1 h_2 U[e_3, e_3, f_1, f_1] - h_1^2 U[e_3, e_3, f_1, f_2] + \frac{1}{2} h_1^2 h_2 U[e_3, l_2, l_2, f_1] - \frac{1}{2} h_1^3 U[e_3, l_2, l_2, f_2] - \frac{1}{2} h_1^2 h_2 U[e_3, l_3, l_3, f_1] + \frac{1}{2} h_1^3 U[e_3, l_3, l_3, f_2] + h_1 h_2 U[e_2, e_3, l_2, f_1, f_1] - h_1^2 U[e_2, e_3, l_2, f_1, f_2] - h_1 h_2 U[e_3, e_3, l_3, f_1, f_1] + h_1^2 U[e_3, e_3, l_3, f_1, f_2] + \frac{1}{2} h_2 U[e_2, e_2, e_3, f_1, f_1, f_1] - \frac{1}{2} h_1 U[e_2, e_2, e_3, f_1, f_1, f_2] - \frac{1}{2} h_2 U[e_3, e_3, e_3, f_1, f_1, f_1] + \frac{1}{2} h_1 U[e_3, e_3, e_3, f_1, f_1, f_2]$$

```
With[{n = 3}, Simp[R1,2[n] ** R1,3[n] ** R2,3[n] - R2,3[n] ** R1,3[n] ** R1,2[n]] // ToDegree[n]]
```

```
0
```

The "Internal Multiplication" and Meta-Associativity

```
m[i_, j_, k_][ℰ_] := Simp[ℰ /. {
    u_U :> UU @@ Join[DeleteCases[u, x_i|j], U @@ Cases[u, x_i :> x_k], U @@ Cases[u, x_j :> x_k]],
    h_i|j → h_k
   }]
```

```
UU[e1, l4, f2]
```

```
U[e1, l4, f2]
```

```
UU[e1, l4, f2] // m[1, 2, 3]
```

```
U[e3, l4, f3]
```

```
UU[e1, l4, f2] // m[2, 1, 3]
```

```
h3 U[l4] + U[e3, l4, f3]
```

```
Union@Table[
  (u // m[1, 2, 1] // m[1, 3, 1]) - (u // m[2, 3, 2] // m[1, 2, 1]),
  {u, UProducts[{e, l, f}, 4, {3}]}
 ]
```

```
{0}
```

The Invariant of the Trefoil



```
With[{n = 2}, R4,1[n] ** R2,5[n] ** R6,3[n] // ToDegree[n]]
```

$$U[] + h_4\,U[l_1] + h_6\,U[l_3] + h_2\,U[l_5] + \left(1 - \frac{h_4}{2}\right) U[e_1, f_4] + \left(1 - \frac{h_6}{2}\right) U[e_3, f_6] + \left(1 - \frac{h_2}{2}\right) U[e_5, f_2] + \frac{1}{2} h_4^2\,U[l_1, l_1] + h_4 h_6\,U[l_1, l_3] + h_2 h_4\,U[l_1, l_5] + \frac{1}{2} h_6^2\,U[l_3, l_3] + h_2 h_6\,U[l_3, l_5] + \frac{1}{2} h_2^2\,U[l_5, l_5] + h_4\,U[e_1, l_1, f_4] + h_6\,U[e_1, l_3, f_4] + h_2\,U[e_1, l_5, f_4] + h_4\,U[e_3, l_1, f_6] + h_6\,U[e_3, l_3, f_6] + h_2\,U[e_3, l_5, f_6] + h_4\,U[e_5, l_1, f_2] + h_6\,U[e_5, l_3, f_2] + h_2\,U[e_5, l_5, f_2] + \frac{1}{2} U[e_1, e_1, f_4, f_4] + U[e_1, e_3, f_4, f_6] + U[e_1, e_5, f_2, f_4] + \frac{1}{2} U[e_3, e_3, f_6, f_6] + U[e_3, e_5, f_2, f_6] + \frac{1}{2} U[e_5, e_5, f_2, f_2]$$

```
With[{n = 2},
 R4,1[n] ** R2,5[n] ** R6,3[n] // ToDegree[n] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // m[1, 5, 1] //
  m[1, 6, 1]]
```

$$\left(1 + h_1 - \frac{h_1^2}{2}\right) U[] + \left(3\,h_1 + 3\,h_1^2\right) U[l_1] + \left(3 - \frac{3\,h_1}{2}\right) U[e_1, f_1] + \frac{9}{2} h_1^2\,U[l_1, l_1] + 9\,h_1\,U[e_1, l_1, f_1] + \frac{9}{2} U[e_1, e_1, f_1, f_1]$$

```
With[{n = 3},
 R4,1[n] ** R2,5[n] ** R6,3[n] // ToDegree[n] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // m[1, 5, 1] //
  m[1, 6, 1]]
```

$$\left(1 + h_1 - \frac{h_1^2}{2} - \frac{5\,h_1^3}{6}\right) U[] + \left(3\,h_1 + 3\,h_1^2 - \frac{3\,h_1^3}{2}\right) U[l_1] + \left(3 - \frac{3\,h_1}{2} - \frac{3\,h_1^2}{2}\right) U[e_1, f_1] + \left(\frac{9\,h_1^2}{2} + \frac{9\,h_1^3}{2}\right) U[l_1, l_1] + \left(9\,h_1 - \frac{9\,h_1^2}{2}\right) U[e_1, l_1, f_1] + \frac{9}{2} h_1^3\,U[l_1, l_1, l_1] + \left(\frac{9}{2} - 9\,h_1\right) U[e_1, e_1, f_1, f_1] + \frac{27}{2} h_1^2\,U[e_1, l_1, l_1, f_1] + \frac{27}{2} h_1\,U[e_1, e_1, l_1, f_1, f_1] + \frac{9}{2} U[e_1, e_1, e_1, f_1, f_1, f_1]$$

Ordering Symbols

```
O[n_, poly_, specs___] := Module[{vs, us},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) :> (l /. x_i_ :> xs));
  Total[CoefficientRules[Normal@Series[poly, {ħ, 0, n}], vs] /. (p_ → c_) :> c UU @@ (us^p)]]
```

```
TimeUsed[]
```

22.767